

NAG Fortran Library Routine Document

F01ADF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F01ADF calculates the approximate inverse of a real symmetric positive-definite matrix, using a Cholesky factorization.

2 Specification

```
SUBROUTINE F01ADF(N, A, IA, IFAIL)
INTEGER          N, IA, IFAIL
real          A(IA,*)
```

3 Description

To compute the inverse X of a real symmetric positive-definite matrix A , this routine first computes a Cholesky factorization of A as $A = LL^T$, where L is lower triangular. It then computes L^{-1} and finally forms X as the product $(L^{-1})^T L^{-1}$.

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer-Verlag

5 Parameters

- 1: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 2: A(IA,*) – **real** array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the upper triangle of the n by n positive-definite symmetric matrix A . The elements of the array below the diagonal need not be set.
On exit: the lower triangle of the inverse matrix X is stored in the elements of the array below the diagonal, in rows 2 to $n + 1$; x_{ij} is stored in $A(i + 1, j)$ for $i \geq j$. The upper triangle of the original matrix is unchanged.
- 3: IA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F01ADF is called.
Constraint: $IA \geq N + 1$.
- 4: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

The matrix A is not positive-definite, possibly due to rounding errors.

IFAIL = 2

On entry, $N < 0$,
or $IA < N + 1$.

7 Accuracy

The accuracy of the computed inverse depends on the conditioning of the original matrix. For a detailed error analysis see page 39 of Wilkinson and Reinsch (1971).

8 Further Comments

The time taken by the routine is approximately proportional to n^3 .

9 Example

To find the inverse of the 4 by 4 matrix:

$$\begin{pmatrix} 5 & 7 & 6 & 5 \\ 7 & 10 & 8 & 7 \\ 6 & 8 & 10 & 9 \\ 5 & 7 & 9 & 10 \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F01ADF Example Program Text
*      Mark 15 Revised.  NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NMAX, IA
      PARAMETER       (NMAX=8,IA=NMAX+1)
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, J, N
*      .. Local Arrays ..
      real            A(IA,NMAX)
*      .. External Subroutines ..
      EXTERNAL        F01ADF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F01ADF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
```

```

      READ (NIN,*) N
      WRITE (NOUT,*)
      IF (N.GE.0 .AND. N.LE.NMAX) THEN
        READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
        IFAIL = 0
*
        CALL F01ADF(N,A,IA,IFAIL)
*
        WRITE (NOUT,*) 'Lower triangle of inverse'
        DO 20 I = 1, N
          WRITE (NOUT,99998) (A(I+1,J),J=1,I)
20      CONTINUE
      ELSE
        WRITE (NOUT,99999) 'N is out of range: N = ', N
      END IF
      STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,8F9.4)
      END

```

9.2 Program Data

F01ADF Example Program Data

```

4
5.   7.   6.   5.
7.  10.   8.   7.
6.   8.  10.   9.
5.   7.   9.  10.

```

9.3 Program Results

F01ADF Example Program Results

```

Lower triangle of inverse
 68.0000
-41.0000  25.0000
-17.0000  10.0000   5.0000
 10.0000 -6.0000  -3.0000   2.0000

```
